cap log close
log using "2_output\furtherResults.log", replace
* Other Results	Mentioned in the Paper
	* Section 3.1 - Discussion of Table 1 
		** (1): "We observe 41.1% [resp. 33.0%] of subjects in the morning shift allocating more to the other child than to themselves when the recipient is in the in-group [out-group]. The corresponding numbers for the afternoon shift are 45.3% [42.7%]."
			ta Game12_Giving_In if !shift // Giving 3 or more toys amounts to 39.29% + 1.79% of children in the morning shift, thus in total 39.29 + 1.79 = 41.08
			ta Game12_Giving_Out if !shift // Giving 3 or more toys amounts to 33.04% of children in the morning shift
			ta Game12_Giving_In if shift // Giving 3 or more toys amounts to 41.81 + 3.02 + 0.43 = 45.26% of children in the afternoon shift
			ta Game12_Giving_Out if shift // Giving 3 or more toys amounts to 41.38 + 1.29 = 42.67% of children in the morning shift

		** (2): "In the morning shift, 28.1% of subjects favor the child from the other shift over the child from their own shift; in the afternoon, it is 25.9% of subjects."
			ta Game3_Giving_Out if !shift // 25.45 + 2.68 = 28.13 of subjects in the morning shift give 3 or more toys to a child from the other shift
			ta Game3_Giving_Out if shift // 23.28 + 1.72 + 0.86 = 25.86 of subjects in the afternoon shift give 3 or more toys to a child from the other shift

		** Footnote 11 / Note to Figure A.2: "Applying a Kolmogorov-Smirnov test we do not see any significant distributional differences between morning and afternoon shift for any of the three tasks." (Footnote 11)
			ksmirnov Game12_Giving_In, by(shift)
			ksmirnov Game12_Giving_Out, by(shift)
			ksmirnov Game3_Giving_Out, by(shift)


	* Section 3.3 - Discussion of Table 3 
		** (1) and Footnote 12. "In fact, examining the relationship between Palestinian roots and out-group discrimination directly, we find that Jordanian children with Palestinian roots do not discriminate at all in favor of their in-group (i.e., fellow Jordanians)" 
			reg SharingGameDiscrimination PalestinianRoots if !shift, cluster(clustering) // Coefficient for PalestinianRoots shows the difference between Jordanian children with Palestinian roots and those without; _cons shows the level of discrimination of those children with Palestinian roots
			lincom _b[_cons] + _b[PalestinianRoots] // Thus, adding the intercept to the coefficient for PalestinianRoots, we get the level of discrimination for the Jordanian children with Palestinian Roots (.147); it is not significantly different from zero
			reg SharingGameDiscrimination i.b1.PalestinianRoots if !shift, cluster(clustering) // Alternatively, after recoding the "grouping" variable PalestinianRoots, _cons shows the coefficient/intercept for children with Palestinian roots: It's the same as the linear combination above (.147), and it's not significantly different from zero
			ttest SharingGameDiscrimination = 0 if !shift & PalestinianRoots == 1 // we could of course also use a t-test; same estimate (.147), but this does not account for possible correlation of errors at the class level
			
		** (2) "While parental narratives are jointly significant for both populations (F-test, p < 0.08 for the morning shift and p < 0.01 for the afternoon shift), they are particularly important for the Syrian population."
		** AND
		** Footnote 16: A potential correlation between these two variables, or with the time spent at the school variable, might raise issues of multicollinearity. Yet, these correlations are lower than one might expect (pair-wise Pearson correlation coefficients never exceed 0.31), and – for each of the variables that we use – the variation that is unexplained by the remaining variables exceeds commonly used thresholds to indicate multicollinearity.
			
			* Code copied from Tables3andA4.do: 
			reg SharingGameDiscrimination TimeGoingToThisSchool c.FriendsFromOutGroupAtSchool##c.NumFriendsOutGroup SyrianRelatives PalestinianRoots ParAtt_Tragedy ParAtt_JordanDoneEnough ParAtt_RentalPrice ParAtt_Jobs childMale nChildren parentMale  governorateIncome i.b0.parentEducation parentsHaveSmartphone parentsHaveBooks if !shift, rob cluster(clustering) // alternatively, restore this results using "est restore full_m_soc"
			test ParAtt_Tragedy ParAtt_JordanDoneEnough ParAtt_RentalPrice ParAtt_Jobs // ad (2): F-test shows a p-value of 0.0797 < 0.08 for the morning shift
			vif // ad Footnote 16: all variance inflation factors (the reciprocals of the variation unexplained by the remaining variables) are below 3; 3 and 5 are commonly used thresholds to indicate issues of multicollinearity
			
			reg SharingGameDiscrimination TimeGoingToThisSchool c.FriendsFromOutGroupAtSchool##c.NumFriendsOutGroup JordanianRelatives OriginToAmman TimeInJordan  ParAtt_Tragedy ParAtt_JordanDoneEnough ParAtt_RentalPrice ParAtt_Jobs childMale nChildren parentMale governorateIncome i.b0.parentEducation parentsHaveSmartphone parentsHaveBooks if shift & !TimeInJordan_long, rob cluster(clustering) // alternatively, restore this results using "est restore full_a_soc"
			test ParAtt_Tragedy ParAtt_JordanDoneEnough ParAtt_RentalPrice ParAtt_Jobs // ad (2): F-test shows a p-value of 0.0043 < 0.01 for the afternoon shift
			vif // ad Footnote 16: all variance inflation factors (the reciprocals of the variation unexplained by the remaining variables) are below the 3 and 5 are commonly used thresholds to indicate issues of multicollinearity
			
			pwcorr  TimeGoingToThisSchool  FriendsFromOutGroupAtSchool NumFriendsOutGroup, sig // all pairwise corrlations below .31
			pwcorr  TimeGoingToThisSchool  FriendsFromOutGroupAtSchool NumFriendsOutGroup if shift, sig // all pairwise correlations below .31
			pwcorr  TimeGoingToThisSchool  FriendsFromOutGroupAtSchool NumFriendsOutGroup if !shift, sig // all pairwise correlations below .31

 log close
 * Have a nice day!